

JUEGOS DE ORDENADOR

MISTERIO

... COMMODORE 64 ... VIC 20 ... APPLE ... SPECTRUM ... DRAGÓN ... MSX ...



212
Σ

MISTERIO

Jenny Tyler y Chris Oxlade

Ediciones Generales Anaya

Contenido

- 2 Acerca de este libro
- 3 La torre del terror
- 4 La pirámide de las calaveras
- 6 Combate de monstruos
- 8 Tiburón
- 10 Las brujas voladoras
- 12 Microenigma
- 17 Respuestas a las cuestiones planteadas

Acerca de este libro

Los programas de este libro están escritos en una versión estándar de BASIC, uno de los lenguajes de ordenador, y hay líneas de conversión para la mayoría de los principales tipos de ordenadores personales. Busca a la izquierda del listado del programa el símbolo que representa a tu ordenador y después mira en la lista de cambios los necesarios para la versión correcta de esa línea. Los símbolos para los distintos ordenadores son los siguientes:

- ▲ Commodore 64 y VIC 20
- Spectrum
- Apple
- ★ DRAGON
- MSX

Acerca de los juegos

Los juegos de este libro son muy sencillos. Se han creado con la intención de ayudarte a acostumbrarte a tu ordenador y al lenguaje BASIC, tecleando los programas, depurándolos y comprendiendo cómo funcionan. Los programas no tienen gráficos o sonido porque varían mucho de un ordenador a otro, pero puedes intentar añadirlos.

Puedes cambiar y adaptar los juegos a tu gusto. Detrás de cada listado de programa se sugieren maneras de hacerlo, pero también puedes experimentar con tus propias ideas. De esta manera, puedes usar los juegos de este libro como base para conseguir juegos más largos y más complicados hechos por ti mismo.

Teclear y ejecutar los programas

Recuerda que puede resultar difícil y llevar mucho tiempo teclear correctamente los programas, incluso si éstos son cortos. Comprueba cada línea que introduzcas, ya que es fácil cometer errores aunque tengas bastante experiencia.

Cuando hayas introducido la totalidad del programa, compruébalo otra vez, asegurándote de que no te has saltado ninguna línea, espacios o signos de puntuación.

Para empezar el juego, teclea RUN. Antes de empezar, lee la introducción del juego para tener una idea de lo que se supone que éste debe hacer. Si el programa no funciona correctamente, es bastante probable que haya un error en cualquier sitio; por eso, lista el programa y compruébalo de nuevo.

Al acabar el juego, puede que el ordenador te pregunte si quieres jugar otra vez o, por el contrario, decir algo como BREAK in 200, en cuyo caso debes teclear RUN para jugar otra vez.

Cambiar la velocidad

Algunos juegos dependen de la velocidad de tus reacciones y del ordenador. Puede ser necesario ajustar la velocidad. Encontrarás instrucciones para hacerlo a continuación del listado del programa.

Microenigma ha sido escrito por Les Howarth y
Combate de monstruos por Adrian Hall

Ilustrado por Rob McCaig
Editado por Jenny Tyler
Diseñado por Roger Priddy

Queda prohibida la reproducción total o parcial de la presente obra bajo cualquiera de sus formas, gráficas o audiovisuales sin la autorización previa y escrita del editor, excepto citas en revistas, diarios o libros, siempre que se mencione la procedencia de las mismas.

Título original: *Weird Computer Games*

© 1984 Usborne Publishing Ltd.

© 1985, de la edición española.

Ediciones Generales Anaya

Villafranca, 22. 28028 Madrid

ISBN: 84-7525-310-5

Depósito legal: M. 39.274. 1985

Impreso por Melsa

Carretera de Fuenlabrada a Pinto, km. 21,800

Pinto (Madrid)

Impreso en España - Printed in Spain

La torre del terror



Tienes la garganta seca, te tiemblan las piernas y te palpita el corazón; has entrado en la Torre del Terror...

Pulsa A para moverte por las habitaciones. ¡¡Agh!! ¡Hay un esqueleto, después un fantasma y un hombre sin cabeza! Con cada nuevo susto, tu pulso aumenta desmesuradamente.

¿Vas a seguir (A), o retroceder (D) para recobrarte un poco? Mira la hora; solamente tienes hasta la medianoche para alcanzar la cima de la Torre y el codiciado Tesoro. También debes estar atento a tu pulso. La locura llegará cuando alcance 150 pulsaciones. ¡Entonces nada podrá impedir que te tires por la ventana!

```

10 GOSUB 470
20 LET R$="BUENA SUERTE"
30 LET RM=0
40 LET H=9:LET M=INT(RND(1)*10)+10
50 LET P=50
60 CLS:PRINT:PRINT
70 PRINT "LA TORRE DEL TERROR"
80 PRINT "*****"
90 PRINT:PRINT R$
100 LET R$="":LET FL=INT(RM/5)
110 LET R=RM-FL*5+1
120 PRINT:PRINT "ESTAS EN"
130 IF FL=0 THEN PRINT "LA PLANTA BAJA"
140 IF FL=6 THEN PRINT "LA PLANTA SUPERIOR"
150 IF FL>0 AND FL<6 THEN PRINT "EL PISO ";FL
160 PRINT "EN LA HABITACION ";R
170 PRINT:PRINT "LA HORA ES ";H;". ";M; " PM"
180 PRINT:PRINT "TU PULSO ES ";P
190 LET GF=0
200 IF RM=30 THEN GOTO 350
210 LET TR=INT(RND(1)*9+1)
220 IF RND(1)>.6 THEN GOSUB 420
230 PRINT:PRINT "HACIA ATRAS O HACIA DELANTE (A/D) ?"
240 LET I$=INKEY$
250 IF I$<>"D" AND I$<>"A" THEN GOTO 240
260 IF I$="D" THEN GOSUB 400
270 IF I$="A" THEN LET RM=RM-1:LET P=P-5
280 IF RM=-1 THEN LET RM=0
290 LET M=M+INT(RND(1)*3+1):IF M>59 THEN LET M=M-60:LET H=H+1
300 IF H=12 THEN GOTO 360
310 IF P>150 THEN GOTO 380
320 IF P<40 THEN LET P=40
330 IF FL=TR AND RND(1)>.5 THEN GOSUB 520
340 GOTO 60
350 PRINT "BIEN HECHO ":STOP
360 PRINT:PRINT "ES MEDIANOCHES!"
370 PRINT:PRINT "DEMASIADO TARDE!":STOP
380 PRINT "TE HAS VUELTO LOCO Y"
390 PRINT "TE HAS TIRADO POR UNA VENTANA!":STOP
400 IF GF=1 THEN LET P=P+5:LET R$="AAAAHHH!"
410 LET P=P-1:LET RM=RM+1:RETURN
420 LET TY=INT(RND(1)*3+1)
430 LET W$=G$(TY)
440 LET S=INT(RND(1)*5)+FL+TY*2
450 PRINT:PRINT "DELANTE DE TI VES UN ";W$
460 LET GF=1:RETURN
470 DIM G$(3)
480 LET G$(1)="ESQUELETO"
490 LET G$(2)="FANTASMA"
500 LET G$(3)="HOMBRE SIN CABEZA"
510 RETURN
520 LET R$="CAES POR UNA TRAMPA":LET RM=RM-5:LET P=P+10:RETURN
    
```

Intenta cambiar el límite del pulso.

¿Puedes también averiguar cómo añadir más habitaciones?

¿Serías capaz de añadir más desgracias y cambiar sus efectos en el pulso?

¿Qué es el Tesoro de la Torre y cómo lo conseguirás...? Quizá puedas añadir tus ideas al programa.

Cómo funciona

10: Va a la sección de inicialización para leer los datos.

30: RM es la habitación donde estás. 40-50: Define la hora de comienzo y el pulso.

90: Imprime cualquier comentario almacenado en R\$.

100-160: Calcula el piso y la habitación en que estás con el valor de RM e imprime esta información.

170-180: Imprime otras informaciones.

190: GF es un switch que vale 1 cuando hay algo desagradable delante.

200: Comprueba si has llegado a la última habitación.

210: Selecciona una habitación para la trampilla.

220: Elige aleatoriamente si hay algo delante. Si es así, salta a una subrutina.

230-280: Captura la respuesta del jugador y actúa en consecuencia.

290: Incrementa el tiempo.

300: Comprueba si es medianoche.

310: Comprueba si el pulso es demasiado alto.

330: Comprueba si estás en una habitación con trampilla.

400-410: Rutina de seguir adelante. Incrementa el pulso si hay algo desagradable.

420-460: Selecciona el tipo de susto y su incremento en el pulso.

470-510: Lee los datos del comienzo del juego (inicialización).

520: Subrutina de la trampilla.

Líneas de conversión

□ 40,210,220,290,330,420,440 Reemplazar RND(1) por RND

■ 40,210,220,290,330,420,440 Reemplazar RND(1) por RND(0)

★ 40,210,220,290,330,420,440 Reemplazar RND(1) por RND(-TIME)

▲ 60 PRINT CHR\$(147):PRINT:PRINT

● 60 HOME:PRINT:PRINT

▲ 240 GET I\$

● 240 I\$="":IF PEEK(-16384)>127 THEN GET I\$


□ 470 DIM G\$(15)

La pirámide de las calaveras

Una mente sin un cuerpo: es todo lo que eres desde aquel terrible accidente, hace tantos años, durante la construcción de la Pirámide.

Toda tu energía mental se dedica a satisfacer las demandas del Señor Diabólico de los Espíritus. Todos los días introduces un número en tu mente (¡si pudieras deshacerte de él!). Debes tirar calaveras desde la Pirámide sobre las piedras numeradas de debajo intentando sumar ese número.

Pulsa 1, 2, 3 ó 4 para tirar una calavera. Debes usar cinco cada vez; ni más, ni menos. Consigue el número correcto y te librarás de la Maldición de la Pirámide.



Incrementa el valor de SP en la línea 10 para aumentar la velocidad del programa.

```
10 LET TS=0:LET I$="0":LET SP=3
20 LET N=INT(RND(1)*30)+10
30 CLS:PRINT
40 PRINT "PIRAMIDE DE LAS CALAVERAS"
50 PRINT
60 PRINT "TIENES QUE "
70 PRINT "CONSEGUIR : ";N:PRINT
80 LET K=3000:GOSUB 540
90 CLS:LET Y=2
100 FOR I=1 TO 4
110 LET A$=STR$(I):LET X=6+2*I
120 GOSUB 560
130 NEXT I
140 LET A$="^":FOR I=1 TO 5
150 LET Y=2+2*I
160 FOR J=1 TO I+3
170 LET X=7-I+2*J
180 GOSUB 560
190 NEXT J
200 NEXT I
210 LET Y=14
220 FOR I=1 TO 9:LET A$=STR$(I)
230 LET X=1+2*I:GOSUB 560
240 NEXT I
250 LET S=0
260 FOR I=1 TO 5
270 LET A$="ELIGE FILA"
280 LET X=0:LET Y=0:GOSUB 560
290 LET I$=INKEY$
300 IF I$="" THEN GOTO 290
310 LET R=VAL(I$)
320 IF R<1 OR R>4 THEN GOTO 290
330 LET Y=3:LET K=400:LET X=6+2*R
340 LET F=0
```

Cómo funciona el programa

10: Define las variables al principio del juego.

20: Elige el número que debe conseguir el jugador.

30-80: Imprime el mensaje de comienzo y espera para que el jugador pueda leerlo.

90-240: Dibuja la pirámide y los números en su cima y en su base.



Puede que seas capaz de sustituir el símbolo de la calavera por un carácter gráfico.

Añade, si puedes, ruidos desagradables según caen las calaveras por La Pirámide.

250: Inicializa a cero el tanteo.

260: Define el número de calaveras.

270-280: Imprime un mensaje en la parte superior izquierda de la pantalla.

290-300: Espera la pulsación de una tecla.

310-320: Comprueba si la tecla pulsada está entre 1 y 4 y en caso contrario, la pide de nuevo.

330: X e Y son las posiciones de la calavera. Esta línea las sitúa en la parte superior de la pirámide.

340: F es un switch que registra si el movimiento anterior era un rebote en el lateral de la pirámide.

Te apuesto a que no
averiguas cómo añadir
más calaveras
cada vez.



```

350 LET A$="0"
360 GOSUB 560
370 GOSUB 540
380 LET A$=" ":GOSUB 560
★ 390 IF F=0 AND Y<13 THEN LET X=X+SGN(RND(1)-.5)
400 LET Y=Y+1:LET F=1-F
410 IF Y<14 THEN GOTO 350
420 LET S=S+(X-1)/2
430 LET X=15:LET Y=0
440 LET A$=STR$(I)+" ":"STR$(S)
450 GOSUB 560
460 NEXT I
470 LET K=1000:GOSUB 540
480 LET D=ABS(S-N):LET TS=TS+D
● 490 CLS:PRINT:PRINT
500 PRINT "TU TANTEO AHORA ES "TS
510 IF D>1 THEN LET K=3000:GOSUB 540:GOTO 20
520 PRINT "Y QUEDAS LIBRE!"
530 STOP
540 FOR T=1 TO K STEP SP:NEXT T
550 RETURN
●★ 560 PRINT TAB(X,Y);A$
570 RETURN

```

350-370: Imprime la calavera y espera.

380: Borra la calavera.

390: Elige la dirección del rebote.

400: Mueve la calavera una fila hacia abajo.

410: Comprueba si ha llegado al final.

420: Calcula el tanteo según la columna donde termina la calavera.

430-450: Imprime el tanteo.

470: Espera.


480-500: Suma la diferencia entre el número requerido y el tanteo, al tanteo total. Borra la pantalla e imprime este último.

510: Si la diferencia es mayor que 1, vuelve a jugar otra vez.

520: Imprime el mensaje de éxito.

540-550: Subrutina para producir una pausa durante una longitud de tiempo que depende de K.


560-570: Subrutina para alterar la posición X,Y del cursor en la pantalla.



Supongo que no
eres lo bastante
listo como para
cambiar los
límites del número
a conseguir.

Líneas de conversión

- 20,390 Reemplazar RND(1) por RND
- 20,390 Reemplazar RND(1) por RND(0)
- ★ 20,390 Reemplazar RND(1) por RND(-TIME)
- 30,90,490 Reemplazar CLS por HOME
- ▲ 30,90,490 Reemplazar CLS por PRINT CHR\$(147)
- 290 I\$="":IF PEEK(-16384))127 THEN GET I\$
- ▲ 290 GET I\$
- 560 PRINT AT Y,X;A\$
- 560 VTAB(Y+1):HTAB(X+1):PRINT A\$
- ▲ 560 PRINT CHR\$(19);:FOR LL=1 TO Y:
PRINT:NEXT:PRINT TAB(X);A\$
- 560 PRINT Y*32+X,A\$
- ★ 560 LOCATE X,Y:PRINT A\$



Combate de monstruos

¡El combate de monstruos es un deporte para lunáticos! Esto no dice mucho a tu favor, el cerebro a cargo de esta masa de huesos y músculos que está a punto de luchar con algunos de los más desagradables monstruos del Universo.

Como cerebro, debes hacer un montón de cálculos rápidos y precisos. Debes calcular el esfuerzo muscular necesario para mantener alejado al monstruo, por ejemplo, y esto implica multiplicar el tipo del monstruo por la distancia que lo separa de ti.

Si te parece demasiado difícil, puedes pulsar el Botón del Pánico (tecla P). Debes entonces averiguar cuánta adrenalina necesita el cuerpo para superar la crisis, dividiendo el incremento de pulsaciones necesario por el suministro de oxígeno. Ten cuidado, no obstante, de no usar demasiado el Botón del Pánico para no hacer que se esfuerce el corazón, lo que podría provocar un infarto.

Para vivir y poder luchar otra vez, debes sobrevivir 12 asaltos contra el monstruo.

Cómo funciona el programa

```
20 LET P=0
30 LET K=3
●▲ 40 CLS
50 LET X=1
60 LET Y=6
70 LET N=-1
75 N=N+1
★■ 80 LET G=INT(RND(1)*Y+X)
★■ 90 LET I=INT(RND(1)*K+K)
100 LET Y=Y+.5
110 LET X=X+.5
120 LET K=K+.5
130 PRINT
140 PRINT
150 PRINT "TIPO DE MONSTRUO : ";
160 PRINT G
170 PRINT
180 PRINT "DISTANCIA : ";
190 PRINT I
200 PRINT
210 PRINT "ESFUERZO MUSCULAR ? ";
220 GOSUB 570
230 IF Z<>I THEN GOTO 320
●▲ 240 CLS
250 PRINT "EL MONSTRUO SE
MANTIENE EN LA BAHIA"
260 IF N<11 THEN GOTO 75
270 PRINT "UFFF !!!!- EL MONSTRUO"
280 PRINT "ESTA CANSADO Y HA IDO A"
290 PRINT "BUSCAR OTRA VICTIMA."
300 PRINT "SOBREVIVES PARA CONTARLO"
310 STOP
●▲ 320 CLS
330 PRINT "ACABAS DE MORIR"
340 PRINT "REVENTADO EN LOS"
350 PRINT "ENORMES BRAZOS DEL MONSTRUO"
360 PRINT
370 PRINT "HAS SOBREVIVIDO ";N;" ASALTOS"
```

20: Inicializa a cero el número de veces que se utiliza el Botón del Pánico.

30-60: Establece los límites superiores del tipo de monstruo y de distancia.

70: Define el número de asaltos.

80: Define la distancia.

100-120: Incrementa los límites superiores de tipo y distancia en 0,5 cada asalto.

210-220: Pide respuesta y salta a la subrutina que la usa.

230: Si la respuesta es incorrecta, salta a la parte del programa donde se imprime el mensaje de fracaso.

250: Imprime un mensaje si has ganado el asalto.

260: Vuelve para otro asalto.

270-310: Mensaje de éxito si sobrevives 12 asaltos.

330-380: Mensaje de fracaso.

Si pierdes siempre, harías mejor bajando la velocidad del programa incrementando el valor de Q en la línea 660.



Disminuye el valor de Q si te resulta fácil.





```

380 STOP
●▲ 390 CLS
★■ 400 LET WX=INT(RND(1)*9+1)
★■ 410 LET WY=INT(RND(1)*9+1)
420 LET U=WX*WY
430 LET P=P+1
440 IF P=4 THEN GOTO 700
450 IF P=3 THEN PRINT "ESTAS VIENDO
LAS ESTRELLAS"
460 PRINT "EL PANICO CONTINUA"
470 PRINT
480 PRINT "INCREMENTO DEL PULSO : ";U
490 PRINT "SUMINISTRO DE OXIGENO = ";WX
500 PRINT
510 PRINT "CANTIDAD DE ADRENALINA ? ";
520 LET Q=100
530 GOSUB 580
540 IF Z<WY THEN GOTO 320
●▲ 550 CLS:IF N<11 THEN GOTO 75
560 GOTO 270
570 LET Q=0
□ 580 LET Z$=""
●▲ 590 LET A$=INKEY$
□ 600 IF A$=CHR$(13) THEN GOTO 680
□ 610 IF A$="P" THEN GOTO 390
620 IF VAL(A$)=0 AND A$<>"0" THEN GOTO 650
□ 630 PRINT A$;
640 LET Z$=Z$+A$
650 LET Q=Q+1
●■ 660 IF Q=500 THEN GOTO 320
670 GOTO 590
680 LET Z=VAL(Z$)
690 RETURN
●▲ 700 CLS
710 PRINT "TE HA DADO UN
ATAQUE AL CORAZON"
720 STOP

```

Para hacerlo más difícil averigua el modo de incrementar el número de asaltos.



400-410: Define el suministro de oxígeno y la cantidad de adrenalina necesaria.

420: Define el incremento de pulsaciones.

430-450: Incrementa la cuenta del Botón del Pánico, imprimiendo aviso después del tercero y acabando el juego después del cuarto.

460-510: Imprime mensajes de pánico y pide respuesta.

520: Define el valor de Q para el comienzo de la cuenta atrás en las líneas 650-660.

530: Salta a la subrutina de respuesta.

540: Si la respuesta es incorrecta, salta al mensaje de fracaso.

550-560: Si es correcta, va al siguiente asalto o al mensaje de éxito.

570-690: SUBROUTINA DE RESPUESTA.

570: Inicializa a cero la cuenta.

580: Define un espacio de memoria para Z\$.

590: Comprueba si se ha pulsado una tecla.

600: Comprueba si la tecla pulsada es RETURN (CHR\$(13)). Si es así, salta a la línea 680.

610: Comprueba si se ha pulsado la tecla P y si es así, salta a la rutina del pánico.

620: Si no se ha pulsado ninguna tecla, salta para incrementar la variable de conteo Q.

630: Si se pulsa otra tecla distinta de RETURN o P, se imprime.

640: Z\$ contiene todas las teclas pulsadas en un asalto.

650-660: Bucle de tiempo. Si Q llega a 1000 antes de que se haya pulsado una tecla, salta al mensaje de fracaso.

670: Vuelve para comprobar la siguiente tecla pulsada.

680: Convierte los números de la cadena Z\$ en valores colocados en Z.

710: Mensaje de fracaso por usar demasiado el Botón del Pánico.

Líneas de conversión

- ▲ 40,240,320,390,550,700 Sustituir CLS por PRINT CHR\$(147)
- 40,240,320,390,550,700 Sustituir CLS por HOME
- 80,90,400,410 Sustituir RND(1) por RND
- 80,90,400,410 Sustituir RND(1) por RND(0)
- ★ 80,90,400,410 Sustituir RND(1) por RND(-TIME)
- 580 LET Z\$=""
- ▲ 590 GET A\$
- 590 A\$="":IF PEEK(-16384)>127 THEN GET A\$
- 595 IF A\$="" THEN GOTO 650
- 597 PAUSE 40
- 600 IF A\$=CHR\$(13) THEN PAUSE 20:GOTO 680
- 610 IF A\$="P" THEN PAUSE 30:GOTO 390
- 630 PRINT A\$;PAUSE 15
- 660 IF Q=250 THEN GOTO 320

Tiburón

¡Mmmmm... las personas están exquisitas! Pero, con el cazador siempre en tu cola, te es difícil acercarte lo suficiente para poder cogerlas.

Tú eres T (de Tiburón) y puedes moverte por donde quieras pulsando las teclas A, Z, N y M. Vamos a ver cuántas P (Personas) puedes comerte antes de que C te cace. Hay un obstáculo (por supuesto). Cada vez que tengas éxito al coger y devorar un delicioso ser humano, te excitarás tanto que no podrás recordar para qué sirve cada tecla.

Cómo funciona el programa

10 DIM M\$(12):DIM K\$(4):GOSUB 660

20 LET S=0:LET T=0:LET G=0

30 LET U=1

40 LET K\$(1)="A":LET K\$(2)="M"

50 LET K\$(3)="Z":LET K\$(4)="N"

60 LET PX=2:LET PY=2

70 GOSUB 600

80 GOSUB 490:GOSUB 440

90 LET NX=PX:LET NY=PY

▲ 100 LET I\$=INKEY\$

110 IF I\$=K\$(1) THEN LET NY=NY-1

120 IF I\$=K\$(2) THEN LET NX=NX+1

130 IF I\$=K\$(3) THEN LET NY=NY+1

140 IF I\$=K\$(4) THEN LET NX=NX-1

150 LET X=NX:LET Y=NY:GOSUB 530

160 IF F=1 THEN GOTO 230

170 LET X=PX:LET Y=PY:LET A\$=" "

180 GOSUB 640

190 LET X=NX:LET Y=NY:LET A\$="T"

200 GOSUB 640

210 LET PX=NX:LET PY=NY

220 IF PX=TX AND PY=TY THEN GOSUB 340

230 IF PX=GX AND PY=GY THEN GOTO 280

★ 240 IF RND(1)>U THEN GOSUB 370

250 LET X=22:LET Y=12:LET A\$=STR\$(T)

260 GOSUB 640

270 LET G=G+1:GOTO 90

280 FOR P=1 TO 2000:NEXT P

10: Define el espacio de memoria para el laberinto. Salta a la subrutina que lee los datos del mismo.

20: Inicializa las variables al comienzo del programa.

30: Define la frecuencia de movimientos del cazador.

40-50: Define las direcciones de movimiento de las teclas.

60: Define la posición de comienzo del tiburón.

70: Salta a la subrutina que imprime el laberinto.

80: Salta a la subrutina que elige las posiciones del cazador y la persona.

90: Coloca al tiburón en la nueva posición.

100-140: Calcula la nueva posición del tiburón según la tecla pulsada.

150: Comprueba que esta nueva posición no esté en un muro.

160: Si es un muro, no se mueve.

170-180: Imprime un espacio en la antigua posición del tiburón.

190-200: Imprime una T en la nueva posición del tiburón.

210: Iguala la posición del tiburón a la nueva posición.

220: Comprueba si la T y la P coinciden y si es así, salta a una subrutina.

230: Comprueba si la T y la C coinciden y si es así, salta al fin del juego.

240: Mueve el cazador.

250-260: Imprime el tanteo.

270: Incrementa el número de movimientos y vuelve para ejecutar el siguiente.

280: Pausa para que puedas ver si te han cazado.

Te reto a que hagas más frecuente el movimiento del cazador.

Líneas de conversión

10 DIM M\$(12,16):DIM K\$(4):GOSUB 660

100 I\$="":IF PEEK(-16348)>127 THEN GET I\$

100 GET I\$

240,390,400,550,560 Cambiar RND(1) por RND

240,390,400,550,560 Cambiar RND(1) por RND(0)

240,390,400,550,560 Cambiar RND(1) por RND(-TIME)

290,600 Cambiar CLS por HOME

290,600 Cambiar CLS por PRINT CHR\$(147):530 LET

F=0:IF M\$(Y,X)<>" " THEN LET F=1


640 PRINT AT Y,X;A\$

640 VTAB(Y):HTAB(X):PRINT A\$

640 PRINT CHR\$(19):FOR LL=1 TO I:PRINT:NEXT:PRINT TAB(X);A\$

640 PRINT Y+32+X,A\$;

640 LOCATE X,Y:PRINT A\$




```

●▲ 290 CLS:PRINT "ACABAN DE CAPTURARTE !"
300 LET S=INT(10000*(T/6)):PRINT "TU PUNTUACION ES : ";S
310 PRINT:PRINT "OTRA PARTIDA ? (S/N) "
320 INPUT A$:IF A$="S" THEN GOTO 20
330 STOP
340 LET T=T+1:LET U=U-.03
350 GOSUB 390:GOSUB 490
360 RETURN
370 LET X=GX:LET Y=GY:LET A$=" "
380 GOSUB 640:GOSUB 440:RETURN
★■ 390 LET K1=INT(RND(1)*4)+1
★■ 400 LET K2=INT(RND(1)*4)+1
410 LET T$=K$(K1)
420 LET K$(K1)=K$(K2):LET K$(K2)=T$
430 RETURN
440 GOSUB 550
450 IF X=TX AND Y=TY THEN GOTO 440
460 LET GX=X:LET GY=Y
470 LET A$="C":GOSUB 640
480 RETURN
490 GOSUB 550
500 LET TX=X:LET TY=Y
510 LET A$="P":GOSUB 640
520 RETURN
□ 530 LET F=0:IF MID$(M$(Y),X,1)<>" " THEN LET F=1
540 RETURN
★■ 550 LET X=INT(RND(1)*4)+2
★■ 560 LET Y=INT(RND(1)*9)+2
570 GOSUB 530
580 IF F=1 THEN GOTO 550
590 RETURN
●▲ 600 CLS:LET X=1
610 FOR Y=1 TO 12:LET A$=M$(Y)
620 GOSUB 640:NEXT Y
630 RETURN
★■●▲ 640 PRINT TAB(X,Y);A$
650 RETURN
660 FOR I=1 TO 12:READ M$(I):NEXT I
670 RETURN
680 DATA "0000000000000000"
690 DATA "0"
700 DATA "0 0 000 00 000 0"
710 DATA "0 0 0"
720 DATA "0 0 0000 0 0 0"
730 DATA "000 0 0 0 0 0"
740 DATA "0 00 0"
750 DATA "0 0 00 0 0 0 0"
760 DATA "0 0 00 0 000 0 0"
770 DATA "0"
780 DATA "0000000000000000"
790 DATA "PERSONAS DEVORADAS : "

```



El tanteo depende del número de personas devoradas y del de movimientos.

290-330: Imprime el mensaje y el tanteo si el tiburón ha sido capturado. Pregunta si quieres jugar otra vez.

340-360: Subrutina que incrementa el tanteo y la frecuencia de movimiento del cazador. Salta a otra subrutina para cambiar la función de las teclas y establecer la nueva posición de la persona.

370-380: Subrutina que borra al cazador de la posición antigua y lo imprime en la nueva.

390-430: Subrutina que cambia la función de las teclas.

440-480: Subrutina para imprimir el cazador, usando otra subrutina para elegir una posición aleatoria. La línea 450 comprueba que en la posición elegida no hay ya una persona.

490-520: Subrutina que imprime la persona, usando otra subrutina para elegir una posición aleatoria.


530-540: Subrutina que comprueba si la posición X,Y coincide con un muro. Si es así, pone F a uno.

550-590: Subrutina que elige una posición aleatoria y comprueba que no coincide con un muro.

600-630: Subrutina que imprime el laberinto.

640-650: Subrutina que imprime en la posición X,Y.

660-790: Subrutina que lee los datos del laberinto.



Estas líneas conforman el laberinto. Puedes cambiarlas dibujando uno nuevo en un papel cuadriculado y copiándolo en el programa, usando O para los muros y espacios para los pasillos.

Las brujas voladoras



Como ayudante en jefe de la Superbruja, tu trabajo consiste en volar en tu escoba recogiendo ingredientes frescos para sus podridas pocimas. Sabrás exactamente lo que tienes que recoger cada vez porque ella transmite sus repugnantes recetas directamente a la pantalla de tu ordenador.


Los ingredientes como P (Patatas de lagarto) y S (Sesos de murciélago) pasan rápidamente por debajo de ti según vuelas. Pulsa cualquier tecla para bajar y recoger alguno de ellos (debes aterrizar justo encima para hacerlo). Ten cuidado, porque alguno de ellos podría ser el Robaescobas, disfrazado. Si aterrizas sobre él, perderás tu escoba. (Afortunadamente, comienzas el juego con tres de repuesto.) Debes recoger todo lo que necesitas antes de llegar a la guarida o la Superbruja no estará contenta contigo...

Cómo funciona el programa

```
10 GOSUB 700:GOSUB 620
20 LET NW=4:LET M=0:LET FM=50
30 GOSUB 530:LET M=NH
40 GOSUB 550
50 LET EF=0:LET FF=0:LET HH=M
60 FOR I=1 TO 4:LET G(I)=0:NEXT I
70 FOR I=1 TO 4:LET A$=P$(I)+":":LET X=1
80 LET Y=1+I:GOSUB 610:NEXT I
90 GOSUB 490:GOSUB 430
100 IF M=FM THEN GOTO 250
110 IF EF=1 THEN GOTO 230
120 IF INKEY$(0)=" " THEN GOTO 90
130 LET HH=M-1:GOSUB 510
140 GOSUB 490:GOSUB 430
150 LET H=HH
160 IF H>0 THEN GOTO 130
170 IF P=17 THEN GOSUB 320
180 IF FF=1 THEN GOTO 240
190 GOSUB 530
200 LET HH=M+1:GOSUB 510:GOSUB 490
210 LET H=HH:IF H=NH THEN GOTO 90
220 GOSUB 430:GOTO 200
230 LET A$="BIEN HECHO ! LA SUPERBRUJA ESTA
    CONTENTA CONTIGO"
240 LET A$="TE QUEDASTE SIN ESCOBAS!":GOTO 260
250 LET A$="DEMASIADO TARDE !"
260 LET X=1:LET Y=15:GOSUB 610
270 STOP
280 LET A$="L":LET X=P:GOSUB 610
290 FOR T=1 TO 1000:NEXT T
300 LET NW=NW-1:IF NW=0 THEN LET FF=1
310 RETURN
320 IF W=1 THEN GOSUB 280:RETURN
330 LET G(R)=G(R)+1
340 LET A$=STR$(G(R)):LET X=23
350 LET Y=1+R:GOSUB 610
360 LET P=19:GOSUB 430
370 LET F=0
380 FOR I=1 TO 4
390 IF G(I)=N(I) THEN LET F=F+1
```



¿Demasiado difícil para ti? Haz el juego más fácil incrementando el valor de FM en la línea 20. Esto produce que aparezcan más objetos entre el comienzo y la Guarida.



Intenta cambiar las letras por símbolos gráficos.

¿Por qué no cambias los ingredientes?

20: Inicializa variables. (Procura averiguar para qué sirve cada una.)

30: H es tu altura sobre el suelo. NH es tu nueva altura después de una zambullida.

50: EF y FF son switches que se ponen a uno si el juego ha terminado. HH es una variable que contiene las alturas intermedias al subir o bajar.

60: G() cuenta cuántas veces has recogido cada tipo de objeto.

70-80: Imprime la receta.

90-120: Mueve el «suelo» y los objetos cruzando la pantalla hasta que pulsas una tecla.

130-160: Mantiene moviéndose el suelo y los objetos, y baja a la bruja hasta el nivel del suelo.

170: Si P=17, la bruja ha aterrizado sobre un objeto.

180: Comprueba los switches del fin de juego.

200-220: Mueve la tierra y los objetos hasta que la bruja alcanza la nueva altura.

230-270: Mensajes de fin.

280-310: Subrutina que muestra una L si el objeto es el Ladrón de escobas disfrazado. Incrementa NW y pone a uno FF si se han usado todas las escobas.

320: Subrutina que comprueba si has aterrizado sobre el Robaescobas.

330-350: Incrementa el número de objetos R recogido e imprime el nuevo total en la pantalla.

370-420: Comprueba si todos los ingredientes para la receta se han recogido.





```

400 NEXT I
410 IF F=4 THEN LET EF=1
420 RETURN
430 LET GD=GD+1:IF GD=6 THEN LET GD=1
440 LET A$=G$(GD):LET X=1:LET Y=15:GOSUB 610
450 LET A$=" ":LET X=P:LET Y=14:GOSUB 610
460 LET P=P+1:IF P=20 THEN GOSUB 550
470 LET X=P:LET A$=C$:GOSUB 610
480 RETURN
490 LET A$="-Y=":LET X=16:LET Y=14-NH
500 GOSUB 610:RETURN
510 LET A$=" ":LET X=16:LET Y=14-N
520 GOSUB 610:RETURN
★ ■ □ 530 LET NH=INT(RND(1)*5+3)
540 RETURN
★ ■ □ 550 LET R=INT(RND(1)*4)+1
★ ■ □ 560 LET V=0:IF RND(1)>.8 THEN LET V=1
□ 570 LET C$=LEFT$(P$(R),1):LET P=1
580 LET N=N+1
590 IF N=FN THEN LET C$="LA GUARIDA"
600 RETURN
★ ■ ▲ □ 610 PRINT TAB(X,Y);A$:RETURN
● ▲ 620 CLS:PRINT:PRINT
630 PRINT "LA POCION DEBE TENER":PRINT
640 FOR I=1 TO 4
★ ■ □ 650 LET N=INT(RND(1)*3)+2:LET N(I)=N
660 PRINT:PRINT N; " ";P$(I)
670 NEXT I:PRINT:PRINT
680 PRINT "PULSA RETURN PARA JUGAR"
● ▲ 690 INPUT X$:CLS:RETURN
□ 700 DIM P$(4):DIM N(4):DIM G(4)
□ 710 DIM G$(5)
720 FOR I=1 TO 4:READ P$(I):NEXT I
730 DATA "PATAS DE LAGARTO","OJOS DE IGUANA","RAICES DE HIEDRA","SESOS DE MURCIELAGO"
740 LET H$=")====>====>====>====>===="
□ 750 FOR I=1 TO 5:LET G$(6-I)=MID$(H$,I,20)
760 NEXT I:LET GD=1
770 RETURN

```

430-480: Imprime una nueva sección de suelo y mueve el objeto hacia la posición P. Si el objeto ha pasado la bruja (es decir P=20), elige un nuevo objeto.

490-500: Imprime la bruja.

510-520: Borra la bruja.

530-540: Elige la nueva altura para la bruja.

550: Elige un objeto.

560: Posibilidad aleatoria de que el objeto sea el Robaescobas.

570: Pone la primera letra del objeto en C\$ y fija la posición P a uno.

580: Incrementa el valor del contador de objeto, M.

590: Si M es igual al máximo número de objetos permitido, C\$ se hace igual a «La Guarida».

610: Subrutina general para imprimir en una determinada posición de pantalla.

620-690: Elige e imprime la receta.

700-770: Inicializa los datos para el comienzo del programa. PS contiene los nombres de los objetos. HS contiene los símbolos del suelo. La línea 750 cambia la posición de los símbolos «>» cada cierto tiempo.

¿Qué hará la Superbruja si fallas? Intenta añadir comentarios y penalidades al juego.

Líneas de conversión

■ 120 IF INKEY\$="" THEN GOTO 90
 ▲ 120 GET I\$:IF I\$="" THEN GOTO 90
 ● 120 I\$="":IF PEEK(-16384)>127 THEN GET I\$
 ● 125 IF I\$="" THEN GOTO 90
 □ 530,550,560,650 Sustituir RND(1) por RND
 ■ 530,550,560,650 Sustituir RND(1) por RND(0)
 ★ 530,550,560,650 Sustituir RND(1) por RND(-TIME)
 □ 570 LET C\$=P\$(R,1):LET P=1
 □ 610 PRINT AT Y,X;A\$:RETURN
 ▲ 610 PRINT CHR\$(19):FOR LL=1 TO Y:PRINT:
 NEXT:PRINT TAB(X);A\$:RETURN

● 610 VTAB(Y):HTAB(X):PRINT A\$:RETURN
 ■ 610 PRINT# Y\$32,X,A\$:RETURN
 ★ 610 LOCATE X,Y:PRINT A\$:RETURN
 ▲ 620,690 Sustituir CLS por PRINT CHR\$(147)
 ● 620,690 Sustituir CLS por HOME
 □ 700 DIM P\$(4,15):DIM N(4):DIM G(4)
 □ 710 DIM G\$(5,24)
 □ 750 FOR I=1 TO 5:LET G\$(6-I)=H\$(I TO I+20)



Microenigma


¿Qué ha ocurrido?

¿Dónde estás?

Todo parece bastante normal, aunque estás un poco mareado. Mejor echar una ojeada y ver si puedes averiguar qué pasa.

(Nota: Intenta hablar con tu ordenador con frases de dos palabras.)

Este es un tipo de juego de miniaventura. Por si nunca has jugado una aventura, te diré que el objetivo es escapar.



Cómo funciona el programa

Líneas 30-360: SECCION PPAL. DEL JUEGO

10 GOSUB 1580

10: Salta a la subrutina que lee los datos.

20 RESTORE:FOR I=1 TO R:READ D\$:NEXT I

30 CLS:PRINT "MICROENIGMA"

40 PRINT "=====

50 PRINT "ESTAS ";D\$

20-50: Coge una descripción de la habitación y la imprime.

60 IF R=20 AND F(9)=0 THEN PRINT "ESTAS FRENTE A UN GATO ENORME"

70 FOR I=1 TO 6

80 IF L(I)=R AND F(I)=0 THEN PRINT "AQUI HAY ";O\$(I);", "

90 NEXT I

70-90: Comprueba si hay algo en ella.

100 PRINT:PRINT "PUEDES IR HACIA ";

110 FOR I=1 TO LEN(R\$(R))

120 PRINT MID\$(R\$(R),I,1);", ";

130 NEXT I

100-130: Averigua en qué direcciones puedes ir y las imprime.

140 PRINT:PRINT "-----"

150 PRINT M\$:LET M\$="QUE ?"

150: Mensaje de error.

160 IF F(16)=1 THEN PRINT "CUENTA ATRAS DE AUTODESTRUCCION EN ; ";L

170 PRINT "QUE HACES AHORA":INPUT Q\$

180 LET V\$="":LET W\$="":LET VB=0:LET OB=0:LET LI=LEN(Q\$)

170-220: Coge tus instrucciones y las divide en dos palabras.

190 FOR I=1 TO LI

200 IF MID\$(Q\$,I,1)=" " AND V\$="" THEN LET V\$=LEFT\$(Q\$,I-1)

210 IF MID\$(Q\$,I+1,1)<>" " AND V\$<>" " THEN LET W\$=RIGHT\$(Q\$,LI-I):LET I=LI

220 NEXT I


```

230 IF V$="" THEN LET V$=Q$
240 FOR I=1 TO V
250 IF V$=H$(I) THEN LET VB=I
260 NEXT I

```

230-260: Comprueba si la primera palabra es un verbo.

```
270 GOSUB 440
```

270: Salta a la subrutina que comprueba la segunda palabra.

```

280 IF VB=0 THEN LET VB=V+1
290 IF V$="" AND OB=0 THEN LET M$="ESO ES UNA TONTERIA"
300 IF V$="" THEN LET M$="NECESITO DOS PALABRAS"
310 IF VB>V AND OB>0 THEN LET M$="NO PUEDES "+Q$
320 IF VB>V AND OB=0 THEN LET M$="NO TIENE SENTIDO"
330 IF VB<V AND OB>0 AND OB<=6 THEN LET M$="NO TIENES "+V$

```

280-330: Define los mensajes necesarios.

```
340 LET L=L-1
```

340: Contador.

```

350 IF VB>14 THEN GOTO 380
360 ON VB GOSUB 480,550,550,550,550,550,550,550,760,790,790,860,920,970
370 GOTO 390
380 ON VB-14 GOSUB 1080,1110,1190,1260,1330,1350,1380,1410,1470,1520,1570

```

350-380: Salta a una subrutina dependiendo del verbo empleado.

```

390 IF F(14)=0 AND L>0 THEN GOTO 20
400 IF L<1 THEN LET M$="HAS CONSUMIDO TODO TU TIEMPO. EL MAXIMINIMIZADOR SE HA AUTODESTRUIDO!"
410 PRINT M$:PRINT "FIN DEL JUEGO"
420 IF L>0 THEN GOSUB 1520
430 STOP

```

390-430: Fin del juego.

Teclea INVENTARIO si quieres una lista de lo que llevas contigo.

```

440 FOR I=1 TO V
450 IF V$=O$(I) THEN LET OB=I
460 NEXT I
470 RETURN

```

440-470: Subrutina que comprueba si la segunda palabra está en la lista de objetos del ordenador.

```

480 PRINT "LLEVAS : "
490 FOR I=1 TO 6
500 IF L(I)=25 THEN PRINT O$(I);", "
510 NEXT I
520 LET M$="":PRINT
530 GOSUB 1560
540 RETURN

```

480-540: Subrutina que te indica lo que llevas contigo.



```

550 LET D=0
560 IF OB=0 THEN LET D=VB-2
570 IF OB>24 THEN LET D=OB-24
580 IF R=19 AND D=6 THEN LET D=4
590 IF R=20 AND D=6 THEN LET D=4
600 IF R=20 AND D=5 THEN LET D=3
610 IF F(9)=0 AND R=20 AND D=4 THEN LET M$="EL GATO NO TE DEJARA":RETURN
620 IF R=2 AND L(1)=25 AND D=3 THEN LET M$=M$:RETURN
630 IF R=7 AND F(3)=0 THEN LET M$="UN RATON ENFADADO CIERRA TU CAMINO":RETURN
640 LET F(13)=0:LET RL=LEN(R$(R))
650 FOR I=1 TO RL
660 LET U$=MID$(R$(R),I,1)
670 IF (U$="N" AND D=1 AND F(13)=0) THEN LET R=R-6:LET F(13)=1
680 IF (U$="S" AND D=2 AND F(13)=0) THEN LET R=R+6:LET F(13)=1
690 IF (U$="O" AND D=3 AND F(13)=0) THEN LET R=R-1:LET F(13)=1
700 IF (U$="E" AND D=4 AND F(13)=0) THEN LET R=R+1:LET F(13)=1
710 NEXT I

```

550-750: Subrutina que procesa tus instrucciones para saber en qué dirección quieres ir.

Si usas un VIC 20, necesitarás la memoria extra para este juego.


```

720 LET M$="OK"
730 IF F(13)=0 THEN LET M$="NO PUEDES IR POR ESE CAMINO"
740 IF B<1 THEN LET M$="ADONDE VAS?"
750 RETURN

```

¿Has jugado unas cuantas veces y piensas que no serás capaz de ganar? Puedes darte más tiempo, si quieres, poniendo un valor más alto para L en la mitad de la línea 1910.



```

760 LET M$="ESTAS PRACTICANDO PARA LAS OLIMPIADAS?"
770 IF F(10)=1 AND (R=9 OR R=3) THEN LET M$="QUIERES SALTAR DEMASIADO LEJOS"
780 RETURN
□ 790 IF OB=8 THEN LET M$=V$+" EL "+W$+M2$:RETURN
800 IF OB>6 THEN LET M$="NO PUEDO COGER "+W$:RETURN
810 IF L(OB)<>R THEN LET M$="NO ESTA AQUI"
820 IF F(OB)<>0 THEN LET M$="QUE "+W$+"?"
830 IF L(OB)=25 THEN LET M$="YA LO TIENES"
□ 840 IF OB>0 AND L(OB)=R AND F(OB)=0 THEN LET L(OB)=25:LET M$=M3$+W$
850 RETURN
860 IF L(OB)<25 THEN RETURN
870 LET M$="REALMENTE NO!":IF OB<>1 THEN RETURN
880 PRINT "DONDE PONGO LA LLAVE"
890 INPUT W$:IF W$="" THEN GOSUB 440
□ 900 IF (OB=8 OR W$="EN "+O$(8)) THEN LET M$=M4$:LET F(15)=1:LET L(1)=R
910 RETURN
□ 920 IF F(11)=0 AND L(7)=25 THEN LET F(11)=1:LET F(3)=0:LET L(3)=0:LET M$=M5$
□ 930 IF F(11)=1 AND L(7)=25 AND R=20 THEN LET F(9)=1:LET M$=M$+M6$
940 IF R=21 AND OB=24 THEN LET F(4)=0:LET M$="HAY UN MONTON DE POLVO"
950 IF OB=16 THEN GOSUB 970
960 RETURN
970 LET M$="NADA INTERESANTE"
980 IF OB=16 AND R=12 THEN LET F(2)=0:LET M$="ES UN MICRO - VCR"
990 IF OB=24 AND R=21 THEN LET M$="ALGO DENTRO"
1000 IF OB=19 AND R=24 THEN GOSUB 1410
□ 1010 IF OB=23 AND R=12 THEN LET M$=M7$
1020 IF OB=7 THEN LET M$="CONTIENE UNA MOSCA MUY GRANDE"
1030 IF OB=9 AND R=20 AND F(9)=0 THEN LET M$="MUERDE Y ARANA!"
1040 IF OB=4 THEN GOSUB 1080
1050 IF OB=1 AND L(1)=25 THEN LET M$="TIENE GRABADO EL NUMERO '111'"
1060 IF OB=6 AND L(6)=25 THEN LET M$="HAY UN GRAN BOTON ROJO"
1070 RETURN
□ 1080 LET M$=M6$:IF OB<>4 THEN RETURN
1090 IF L(OB)=25 THEN LET M$=M6$
1100 RETURN
1110 IF OB<>5 THEN LET M$="NO SE PUEDE ATAR "+W$:RETURN
1120 IF L(5)<25 THEN RETURN
1130 LET W$="":PRINT "DONDE ATO EL HILO?"
1140 INPUT W$
1150 IF W$="" THEN GOSUB 440
□ 1160 LET M$="NO PUEDO ATARLO A "+W$
1170 IF OB=13 AND R=9 THEN LET F(5)=1:LET L(5)=0:LET M$=M8$
1180 RETURN
1190 IF OB=5 AND F(5)=0 THEN LET M$="NO ESTA ATADO A NADA!"
1200 IF OB=5 AND R=9 AND F(5)=1 THEN LET R=8:LET F(10)=0:LET M$="OK":RETURN
1210 IF OB=5 AND R=8 AND F(5)=1 THEN LET R=9:LET F(10)=1:LET M$="OK"
1220 IF OB=13 AND R=9 THEN LET M$="ESTA DEMASIADO RESBALADIZO PARA TREPAR"

```

760-1500: Son subrutinas de los verbos usados en el programa. Busca las líneas donde dice RETURN para ver dónde termina una y empieza otra.

A menos que uses un Spectrum, puedes quitar, si quieres, todos los LET.





```

1230 IF OB=W AND R=19 THEN LET OB=30:GOSUB 550
1240 IF OB=W AND R=20 THEN LET OB=29:GOSUB 550
1250 RETURN
1260 IF OB<>6 OR L(6)<>25 THEN RETURN
1270 LET W$="":PRINT "DONDE APUNTO?"
1280 INPUT W$
1290 IF W$="" THEN GOSUB 440
1300 IF OB=22 THEN LET F(6)=1
1310 LET M$="MUY BIEN"
1320 RETURN
1330 IF OB=3 AND L(OB)=25 THEN LET L(OB)=0:LET M$="MUNCH CHOMP"

```

Este programa no se explica tanto como los otros para no darte demasiadas pistas sobre el juego. Intenta averiguar para qué sirve cada línea después de haber jugado varias veces.



```

1340 RETURN
1350 IF (OB=19 OR OB=17) AND R=24 AND L(1)<25 THEN LET M$=M$:LET L=L-12:LET F(16)=1
1360 IF (OB=19 OR OB=17) AND R=24 AND L(1)=25 THEN LET F(12)=1:LET M$=M$
1370 RETURN
1380 IF OB=3 AND R=7 AND L(OB)=25 THEN LET F(3)=1:LET L(OB)=0:LET B=B+1:LET M$=M$
1390 IF L(OB)=25 THEN LET L(OB)=R:LET M$="HECHO"
1400 RETURN
1410 IF R<24 THEN LET M$="NO HAY DONDE TECLEAR!":RETURN
1420 IF F(12)=0 THEN LET M$="ESTE TERMINAL NO ESTA ACTIVADO":RETURN
1430 LET M$="RESPUESTA DEL TERMINAL :"+W$
1440 IF OB=18 THEN LET M$="PALABRA CLAVE ACEPTADA":LET F(17)=1
1450 IF OB=20 THEN LET M$="IDENTIFICADOR DEL TERMINAL"
1460 RETURN
1470 IF R=14 AND OB=11 AND L(8)=22 THEN LET L(8)=14:LET M$=M$+M$:RETURN
1480 IF R=14 AND OB=11 AND L(8)=14 THEN LET L(8)=22:LET M$=M$+M$
1490 IF F(15)=1 THEN LET L(1)=L(8)
1500 IF OB=21 AND F(6)=1 AND L(6)=25 AND R=18 AND F(17)=1 THEN LET B=B+10:LET M$=M$:LET F(14)=1

```

```

1510 RETURN
1520 LET S=0:FOR I=1 TO 6
1530 IF L(I)=25 THEN LET S=S+1
1540 NEXT I
1550 LET S=S+B:PRINT "TU PUNTUACION = ";S
1560 IF F(14)=0 THEN PRINT "PULSA RETURN PARA CONTINUAR":INPUT @

```

1510-1560: Subrutina de tanteo.

```

1570 RETURN
1580 LET V=24:LET W=31:LET G=8
1590 DIM R$(24):DIM O$(W):DIM H$(V)

```

1570-1590: Inicializa variables y dimensiona tablas.

```

1600 DIM L(6):DIM F(17)
1610 DATA "DENTRO DE LA RATONERA - ESTA MUY OSCURO"
1620 DATA "EN LA RATONERA DE UNA ESQUINA DE LA HABITACION","EN EL BORDE DE UNA MESA ALTA"
1630 DATA "AL FINAL DE UN PASILLO","EN UN ALMACEN","EN LA COCINA"
1640 DATA "MAS AL FONDO DE UN APESTOSO Y OSCURO TUNEL","AL LADO DE LA VIA DEL TREN"
1650 DATA "EN LA BASE DE UN ARBOL ALTO DE PLASTICO EN EL BORDE DE UNA MESA ALTA"
1660 DATA "EN EL EXTERIOR DE UNA CASA EXTRAVAGANTEMENTE PROPORCIONADA QUE TIENE LA PUERTA ABIERTA"
1670 DATA "EN UNA HABITACION AMARILLA","AL LADO DE UN TELEVISOR Y UN CASSETTE"
1680 DATA "AL FINAL DE UN OSCURO TUNEL"
1690 DATA "AL LADO DE UN GRAN INTERRUPTOR CONECTADO A LAS VIAS DEL TREN"
1700 DATA "A LO LARGO DE LA SERPENTEANTE PISTA"
1710 DATA "AL FINAL DE LA LINEA-LA PISTA DESAPARECE A TRAVES DE UN AGUJERO EN LA PARED"
1720 DATA "DEBAJO UNA PARED REPLETA DE PANTALLAS DE VIDEO SOBREDIMENSIONADAS"
1730 DATA "DE PIE SOBRE EL MAXIMINIZADOR"
1740 DATA "EN UNA ESTANTERIA DE APARATOS DESORDENADOS - HAY UN TABURETE AL LADO"

```

1600-1870: Líneas de datos.


```

1750 DATA "SOBRE UN TABURETE BAJO"
1760 DATA "SOBRE UNA CAJA VOLCADA DE COMPONENTES ELECTRONICOS ROTOS"
1770 DATA "EN UN AGUJERO EN LA PARED DEL QUE EMERGE UNA LINEA DE FERROCARRIL"
1780 DATA "EN LA BASE DE UN SILLON GIRATORIO"
1790 DATA "EN UN TERMINAL DE ORDENADOR CON CIERRE DE SEGURIDAD"
1800 DATA 16,12,3,21,5,17,19,14
1810 DATA "INVENTARIO","IR","M","S","O","E","A","D","SALTAR","COGER","TOMAR","PONER","ABRIR"
1820 DATA "EXAMINAR","LEER","ATAR","TREPAS","APUNTA","COMER","ABRIR-CERRADURA","DEJAR","TECLEAR"
1830 DATA "NO","LLAVE","CASSETTE","QUESO","PAPEL","HILO","CONTROL-REMOTO"
1840 DATA "EO","NO","NS","NE","EO","O","S","S","NED","EOD","EO","EO","NEO"
1850 DATA "NO","LLAVE","CASSETTE","QUESO","PAPEL","HILO","CONTROL-REMOTO"
1860 DATA "BOTELLA","TREN","GATO","PUERTA","INTERRUPTOR","TUNEL","ARBOL","AGUJERO","RATON"
1870 DATA "VIDEO","ORDENADOR","ESPECIERO","TERMINAL","III","BOTON","MAXIMINIZADOR"

```

```
1880 DATA "TV","CAJA","NORTE","SUR","DESTE","ESTE","ARRIBA","ABAJO","TABURETE"
```

```
1890 FOR I=1 TO 24:READ D$:NEXT I:FOR I=1 TO 6:READ L(I):NEXT I
```

1880-1900: Lee datos.

```
1900 FOR I=1 TO V:READ M$(I):NEXT I:FOR I=1 TO 24:READ R$(I):NEXT I
```

```

1910 FOR I=1 TO V:READ O$(I):NEXT I:LET F(10)=1:LET F(4)=1:LET F(2)=1
1920 LET R=11:LET B=8:LET L=100:LET M$="DESPIERTAS...":LET MH$="TERMINAL ACTIVO"
1930 LET M1$="NO PUEDES COGER LA LLAVE":LET M2$=" EH? MUY DIVERTIDO!"
1940 LET M3$="TIENES ":LET M4$="BIEN HECHO!":LET M5$="NADA INTERESANTE"
1950 LET M6$="UNA MOSCA VUELA CON UN FUERTE ZUMBIDO":LET M7$="EL TREN CAMINA "
1960 LET M8$="SOLO UNA CAJA CON UNA FOTO CLAVADA":LET M9$="¡ SABOTAJE !!"
1970 LET M10$="ESTA BIEN ATADO.":LET M11$=" Y EL GATO LO PERSIGUE!"
1980 LET M12$="EL RATON HUYE CON EL.":LET M13$="PALABRA CLAVE DEL TERMINAL=" +O$(18)
1990 LET M14$="SE VE Y PARA AQUI":LET M15$="LEJOS Y DENTRO DE UN TUNEL"
2000 LET M16$="FUNCIONA EL RAYO DEL MAXIMINIZADOR. ESTAS VOLVIENDO A TU TAMAÑO NORMAL"
2010 RETURN

```

1910-2010: Define mensajes.

▲ 30 Reemplazar CLS por PRINT CHR\$(147)

● 30 Reemplazar CLS por HOME

Para los usuarios del Spectrum

```

120 PRINT R$(R,I);";";
190 FOR I=1 TO LI-1
200 IF O$(I)="" AND V$="" THEN LET V$=O$(I TO I-1)
210 IF O$(I+1)="" AND V$="" THEN LET W$=O$(I+1 TO I):LET I=LI-1
235 IF LEN(V$)<7 THEN LET V$=V$+" "
235 IF LEN(V$)<7 THEN LET V$=V$+" ":GOTO 235
360 GOSUB 480+(VB=1)+550+(VB=2)+600+(VB=3)+700+(VB=4)+800+(VB=5)+900+(VB=6)+1000+(VB=7)+1100+(VB=8)+1200+(VB=9)+1300+(VB=10)+1400+(VB=11)+1500+(VB=12)+1600+(VB=13)+1700+(VB=14)+1800+(VB=15)+1900+(VB=16)+2000+(VB=17)+2100+(VB=18)+2200+(VB=19)+2300+(VB=20)+2400+(VB=21)+2500+(VB=22)+2600+(VB=23)+2700+(VB=24)+2800+(VB=25)
380 GOSUB 1080+(VB=15)+1100+(VB=16)+1180+(VB=17)+1250+(VB=18)+1320+(VB=19)+1340+(VB=20)+1370+(VB=21)+1400+(VB=22)+1460+(VB=23)+1510+(VB=24)+1560+(VB=25)
440 IF LEN(V$)<14 THEN LET W$=W$+" ":GOTO 440
445 FOR I=1 TO W
660 LET W$=R$(R,I)
900 IF (OB=8 OR W$="EN "+O$(8)+" " THEN LET M$="BIEN HECHO!":LET F(15)=1:LET L(1)=R
1590 DIM R$(24,4):DIM O$(W,14):DIM M$(V,7)
1920 LET R=11:LET B=8:LET L=100:LET M$="DESPIERTAS..."
Lineas 620,790,840,920,930,1010,1080,1170,1350,1360,1380,1470,1480,1500
Sustituir de M1$ a M16$ en estas lineas por el texto de las lineas 1920 a 2000
Ej. 620 IF R=2 AND L(1)=25 AND D=3 THEN LET M$="NO PUEDES COGER LA LLAVE":RETURN
Quitar las lineas 1930 a 2000

```

NOTA IMPORTANTE
para usuarios de Commodore.

Si usas un C64 o un VIC 20, debes quitar los LET de las lineas 400, 1350, 1380 y 1500.

Si usas un C64 debes quitar todos los espacios de las lineas 1500, 1710 y 1860.



Respuestas a las cuestiones planteadas

Aquí están las respuestas a algunas de las cuestiones planteadas en este libro.

La Pirámide de las Calaveras

Para añadir más calaveras, cambia el 5 de la línea 260 por un número más alto.

Los límites del número que habrá que alcanzar, se definen en la línea 20. Tal como está el programa, el número elegido se mantiene entre 10 y 39. Si cambias el 30 por 40, tendrás un número entre 10 y 49. Si también cambias el 10 por 20, tendrás un número entre 20 y 49.

Combate de monstruos

Para incrementar el número de asaltos, cambia el número 11 de las líneas 260 y 550 por un número más alto. (Debe ser el mismo en ambas.)

Si incrementas el número de asaltos, puedes querer más oportunidades de usar el Botón del Pánico. Para hacerlo, cambia el número 4 de la línea 440 por un número más alto.

Tiburón

Para hacer más frecuente el movimiento del cazador, disminuye el valor de U en la línea 30 a, por ejemplo, 0.8.

Las brujas voladoras

Si fallas, el programa salta a la línea 240 ó 250 para imprimir un mensaje. Cambia éstas por lo que tú quieras.



JUEGOS DE ORDENADOR

Cada uno de estos nuevos libros contiene sencillos programas de juegos para utilizar en los ordenadores domésticos de mayor difusión. Junto a cada programa hay explicaciones sobre cómo funciona, así como problemas y sugerencias para introducir cambios que compliquen las aventuras o las hagan más sencillas, según los deseos del usuario. Por medio de estos apasionantes juegos, incluso los que nunca hayan utilizado un ordenador aprenderán con facilidad cómo funciona un programa sencillo, y podrán diseñar alguno por sí mismos. Al final de cada libro aparecen las claves de conversión para diversos ordenadores, así como un glosario de los términos más usuales en BASIC.

Títulos de la serie



12 89

300